ABSTRACT
Data that contain multiple observations per case are called “repeated measures data.” Common examples are medical records, credit card charges, and airline travel history. This tutorial will discuss ways to use the SAS DATA step to handle repeated measures data, in order to generate summary data for reports or for submission to analytic techniques (e.g., regression). We will address both horizontal repeated measures (using arrays) and vertical repeated measures (using RETAIN, FIRST., LAST., OUTPUT, and other SAS statements and options), as well as how to convert horizontally structured data to vertically structured data and vice versa.

INTRODUCTION
Repeated measures data generally take one of two forms: (1) one record per case, with the repeated measures in an array and different variable names for each individual observation, or (2) multiple records per case, with each record containing data for a single episode (such as a medical visit or a credit card transaction). This paper will discuss the advantages and disadvantages of data stored in these two formats, with an emphasis on how to use the SAS DATA step to process these two forms of repeated measures data. We will also see how to convert from one record per case to multiple records per case, and vice versa. This is a basic tutorial that introduces the appropriate SAS tools for processing repeated measures data.

WHY PROCESS REPEATED MEASURES DATA?
By definition, repeated measures data involve multiple events. But most of the time, we are less interested in the individual events than in a summary of some or all of the events. For example, let’s say you had a dataset of all your credit card transactions for the past year and you wanted to know how many times you used your card to buy groceries, and how much you spent altogether on groceries for the entire year. Or, you might be interested in knowing how many times you bought groceries during the month of April, and how much you spent in April on groceries. If we have a record of each individual transaction, we can count the number of occurrences and sum the amounts spent. We can also bracket or filter our counting and summing to include only the part of the data that we are interested in, e.g., only transactions in April. When we are finished, we want to have one output record that gives us the total number of times we charged groceries during the year, the total amount spent during the year on groceries, the total number of times we charged groceries during April, and the total amount spent during April. These four functions—counting, summing, bracketing, and filtering—are the most common methods that we use to process repeated measures data in order to generate a single output record.

Now, let’s take it a step further. Suppose we had a SAS data set on credit card charges, not just for one person, but for everyone who uses a particular kind of credit card (e.g., everyone who has a Visa card issued by Bank of America). We could ask the same questions for each individual, generating one output record per individual. We could then use PROC MEANS to determine the mean number of transactions per credit card, and the mean amount spent on groceries. Or we could use PROC FREQ to see how many had more than 50 transactions during the past year, or PROC PRINT to list the amount spent on groceries by each credit card holder. In fact, repeated measures data, once reduced to one record per case, may serve as input for any SAS PROC.

In this tutorial, we will use a simple dataset to illustrate various ways to process multiple measures data. The same techniques may be used, regardless of the size or complexity of your data.

PROCESSING ARRAY DATA
Continuing with our example of credit card transactions, consider the sample data shown in Figure 1 (this figure shows only part of the data set, which actually contains up to 11 transactions for each of the 4 customer IDs). To simplify our example, we have restricted our sample data set to include only transactions for buying groceries and gasoline for four credit card holders, and only two months’ worth of data (March and April).
Since the data already contain only a single record per case, we can use the SAS ARRAY statement to answer our questions. First we need to use a PROC CONTENTS to determine the maximum number of elements in each array we'll be looking at. The output of the PROC CONTENTS for our sample array data, shown in Figure 2, tells us that each person in our data set may have up to 11 transactions.

First, we use a DATA statement with a KEEP= data set option to keep only the summary variables that we are going to generate, plus an ID for each individual:

```sas
data answer (keep=id alltrans totalamt aprtrans apramt);
set vector;
```

Next, we set up arrays to simplify processing the variables that we will be using for filtering, counting, and summing. Note that since we already know that an individual has a maximum of 11 transactions, we could specify the exact number in the ARRAY statement. For example:

```sas
array trans(11) transdate1-transdate11;
array amount(11) amount1-amount11;
```

This allows us to produce a DATA step that will tell us the total number of grocery transactions and the amount of money spent on groceries, both for the entire period and for the month of April only. We will build up the DATA step code needed to answer these questions in separate pieces, so that we can understand each step in the process.

Figure 2--PROC CONTENTS Output for Sample Array Data Set

This allows us to produce a DATA step that will tell us the total number of grocery transactions and the amount of money spent on groceries, both for the entire period and for the month of April only. We will build up the DATA step code needed to answer these questions in separate pieces, so that we can understand each step in the process.

First, we use a DATA statement with a KEEP= data set option to keep only the summary variables that we are going to generate, plus an ID for each individual:

```sas
data answer (keep=id alltrans totalamt aprtrans apramt);
set vector;
```

Next, we set up arrays to simplify processing the variables that we will be using for filtering, counting, and summing. Note that since we already know that an individual has a maximum of 11 transactions, we could specify the exact number in the ARRAY statement. For example:

```sas
array trans(11) transdate1-transdate11;
array amount(11) amount1-amount11;
```

This allows us to produce a DATA step that will tell us the total number of grocery transactions and the amount of money spent on groceries, both for the entire period and for the month of April only. We will build up the DATA step code needed to answer these questions in separate pieces, so that we can understand each step in the process.
number of elements in the array. Alternatively, we can let SAS determine the number by counting the number of variables that we specify in the definition of the array, by using an asterisk for the array size. These statements illustrate the latter:

```sas
array purpose (*) purpose1-purposell;
array transdate (*) transdate1-transdate11;
array amount (*) amount1-amount11;
```

Now we need to initialize the summary variables that we will use to hold our counts and totals. To add clarity, we also include a LABEL statement to associate a label with each created variable:

```sas
alltrans = 0;
totalamt = 0;
aprtrans = 0;
apramt = 0;
label
  alltrans = 'total number of transactions for groceries'
  totalamt = 'total amount spent on groceries'
  aprtrans = 'number of transactions for groceries in April'
  apramt = 'amount spent on groceries in April';
```

We are now ready to write the code to use the variables in the PURPOSE array as filters, those in the AMOUNT array as counters and summands, and the dates in the TRANSDATE array to bracket the variables we are creating that pertain only to April:

```sas
do i = 1 to 11;
  if purpose {i} = 'groceries' then do;
    alltrans + 1;
    totalamt + amount {i};
    if month (transdate {i}) = 4 then do;
      aprtrans + 1;
      apramt + amount {i};
    end;
  end;
end;
```

This code filters the PURPOSE variables so that we are only counting purchases for groceries, ignoring the gasoline purchases that are also in the data set. Whenever we find a transaction for groceries, we increment ALLTRANS, the variable counting all transactions, and add the amount of the transaction to TOTALAMT, which is going to give us the total amount spent on groceries. Using the TRANSDATE variables, we set up a DO loop that is only used if the transaction occurred in April, and repeat a similar counting and summing process within the DO loop.

Finally, to make sure our created data looks the way we want it to, we add a FORMAT statement:

```sas
format totalamt apramt dollar10.2;
```

This completes the entire DATA step that we need. If we then add

```sas
run;
```

and submit the job, we will create a new data set that contains the variables we are interested in. Figure 3 shows a PROC PRINT output for this data set.
As we have already noted, this new data set can now be used as input to any SAS PROC. Figure 4 shows the output from a PROC MEANS as an example.

**PROCESSING MULTIPLE RECORD DATA**

Many software programs include functions that allow you to process array data in much the same way that we have illustrated above. The particular strength of SAS is in processing repeated measures data where you have a variable number of records per individual. And in contrast to the example above, we do not have to determine the maximum number of observations per individual in order to process multiple records data.

Suppose we have the same data as above, but rather than one record per case with observations listed array-style, we have a separate record for each transaction. Figure 5 shows a portion of the multiple records data set.

**Figure 3--PROC PRINT Output for Summary Data from Example Array Data Set**

<table>
<thead>
<tr>
<th>id</th>
<th>alltrans</th>
<th>totalamt</th>
<th>aprtrans</th>
<th>apramt</th>
</tr>
</thead>
<tbody>
<tr>
<td>1234</td>
<td>7</td>
<td>$511.81</td>
<td>4</td>
<td>$325.58</td>
</tr>
<tr>
<td>1235</td>
<td>6</td>
<td>$503.38</td>
<td>3</td>
<td>$258.25</td>
</tr>
<tr>
<td>1236</td>
<td>5</td>
<td>$377.98</td>
<td>2</td>
<td>$189.89</td>
</tr>
<tr>
<td>1237</td>
<td>8</td>
<td>$645.43</td>
<td>4</td>
<td>$379.74</td>
</tr>
</tbody>
</table>

**Figure 4--PROC MEANS Output for Summary Data from Example Array Data Set**

<table>
<thead>
<tr>
<th>Variable</th>
<th>Label</th>
<th>Mean</th>
<th>Median</th>
</tr>
</thead>
<tbody>
<tr>
<td>alltrans</td>
<td>total number of transactions for groceries</td>
<td>6.50</td>
<td>6.50</td>
</tr>
<tr>
<td>totalamt</td>
<td>total amount spent on groceries</td>
<td>505.85</td>
<td>507.60</td>
</tr>
<tr>
<td>aprtrans</td>
<td>number of transactions for groceries in April</td>
<td>3.25</td>
<td>3.59</td>
</tr>
<tr>
<td>apramt</td>
<td>amount spent on groceries in April</td>
<td>208.37</td>
<td>291.92</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Variable</th>
<th>Label</th>
<th>Minimum</th>
<th>Maximum</th>
<th>N</th>
</tr>
</thead>
<tbody>
<tr>
<td>alltrans</td>
<td>total number of transactions for groceries</td>
<td>5.00</td>
<td>8.00</td>
<td>4</td>
</tr>
<tr>
<td>totalamt</td>
<td>total amount spent on groceries</td>
<td>377.98</td>
<td>645.43</td>
<td>4</td>
</tr>
<tr>
<td>aprtrans</td>
<td>number of transactions for groceries in April</td>
<td>2.00</td>
<td>4.00</td>
<td>4</td>
</tr>
<tr>
<td>apramt</td>
<td>amount spent on groceries in April</td>
<td>189.89</td>
<td>379.74</td>
<td>4</td>
</tr>
</tbody>
</table>

**Figure 5--Part of a Multiple Records Data Set of Repeated Measures**

<table>
<thead>
<tr>
<th>purpose</th>
<th>id</th>
<th>transdate</th>
<th>amount</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1234</td>
<td>03/05/2004</td>
<td>$20.30</td>
</tr>
<tr>
<td>2</td>
<td>1234</td>
<td>03/07/2004</td>
<td>$53.45</td>
</tr>
<tr>
<td>3</td>
<td>1234</td>
<td>03/15/2004</td>
<td>$44.44</td>
</tr>
<tr>
<td>4</td>
<td>1234</td>
<td>03/30/2004</td>
<td>$78.34</td>
</tr>
<tr>
<td>5</td>
<td>1234</td>
<td>03/30/2004</td>
<td>$25.30</td>
</tr>
<tr>
<td>6</td>
<td>1234</td>
<td>04/06/2004</td>
<td>$83.34</td>
</tr>
<tr>
<td>7</td>
<td>1234</td>
<td>04/13/2004</td>
<td>$77.44</td>
</tr>
<tr>
<td>8</td>
<td>1234</td>
<td>04/15/2004</td>
<td>$37.55</td>
</tr>
<tr>
<td>9</td>
<td>1234</td>
<td>04/22/2004</td>
<td>$88.45</td>
</tr>
<tr>
<td>10</td>
<td>1234</td>
<td>04/29/2004</td>
<td>$76.35</td>
</tr>
<tr>
<td>11</td>
<td>1235</td>
<td>03/03/2004</td>
<td>$63.45</td>
</tr>
<tr>
<td>12</td>
<td>1235</td>
<td>03/05/2004</td>
<td>$24.44</td>
</tr>
<tr>
<td>13</td>
<td>1235</td>
<td>03/13/2004</td>
<td>$88.34</td>
</tr>
<tr>
<td>14</td>
<td>1235</td>
<td>03/28/2004</td>
<td>$35.30</td>
</tr>
<tr>
<td>15</td>
<td>1235</td>
<td>03/28/2004</td>
<td>$63.34</td>
</tr>
<tr>
<td>16</td>
<td>1235</td>
<td>04/11/2004</td>
<td>$32.85</td>
</tr>
<tr>
<td>17</td>
<td>1235</td>
<td>04/13/2004</td>
<td>$90.45</td>
</tr>
<tr>
<td>18</td>
<td>1235</td>
<td>04/20/2004</td>
<td>$86.35</td>
</tr>
<tr>
<td>19</td>
<td>1235</td>
<td>04/27/2004</td>
<td>$73.45</td>
</tr>
</tbody>
</table>

Let us look at how we might use these data to generate the same summary variables that we produced with array processing. As before, we will build up the DATA step code gradually, so that each step in the process is clear. First we need to make sure that the data set is sorted by ID:
Now, just as before, we use a DATA statement with a KEEP= data set option to save only the summary variables and an ID for each individual. But now we introduce a new concept into the DATA step, the BY statement, which in this context works with the SET statement. We have paved the way for this usage by the PROC SORT code above, since a BY statement used in a DATA step requires that the data be sorted by the variable(s) specified in the BY statement. Using a BY statement allows us to determine the first and last observation in the “by-group,” i.e., all the records with the same value in the variable specified in the BY statement (and often called the “by-variable”). When you use a BY statement in the DATA step, SAS creates two temporary variables that may be used only in that DATA step. These are FIRST.byvariable and LAST.byvariable. FIRST.byvariable has a value of 1 for the first observation that takes on a new value for the by-variable, 0 otherwise. Similarly, LAST.byvariable has a value of 1 only for the last observation in the by-group, and 0 for all other observations. We will see below that this is a powerful and extremely useful tool in processing these types of data.

```sas
proc sort data=sugi.example
   out=example;
   by id;
run;
```

We also need to use a RETAIN statement for the summary variables we will be creating. RETAIN tells SAS to save the value of the variable even when it goes to the next observation. This is very important in processing repeated measures data in the multiple records format. Without the RETAIN statement, each new observation will start with missing data, rather than your accumulating counts and totals, and the resulting summary variables will not be accurate.

```sas
retain alltrans totalamt aprtrans apramt;
```

Next, we must initialize the summary variables, as we did when processing array data. In this instance, though, since we have multiple records for the same individual, we need to initialize the summary variables only when we encounter a new ID. This is where it becomes critical to identify the first instance of a particular value in the by-variable. We do this with the use of the FIRST. (“first-dot”) feature. Here, since our by-variable is ID, we will use FIRST.id to identify the first time we encounter a new ID:

```sas
if first.id then do;
   alltrans = 0;
   totalamt = 0;
   aprtrans = 0;
   apramt = 0;
end;
```

We now count the number of transactions and accumulate the total amounts spent, in much the same way we did above when the data were in an array. However, with multiple records per case, we do not need to use array processing, since we want to do the same thing for every record we encounter. The SAS code looks like this:

```sas
if purpose = 'groceries' then do;
   alltrans + 1;
   totalamt + amount;
   if month (transdate) = 4 then do;
      aprtrans + 1;
      apramt + amount;
   end;
end;
```

Finally, when we have passed through all the records for the by-group, i.e., when we come to the last record for a given value of the variable ID, we are ready to write out the resulting summary variables. To do this, we use the LAST. (“last-dot”) feature and an explicit OUTPUT statement. In DATA steps that do not include an explicit OUTPUT statement, SAS assumes an implicit OUTPUT statement at the end of the DATA step. The explicit OUTPUT statement tells SAS to create an output record only when we want to do so. Combining the LAST. feature and the
explicit OUTPUT statement tells SAS to create an output record each time we come to the last record in the by-group:

```sql
if last.id then output;
```

Although LABEL and FORMAT statements can go anywhere in a DATA step, it seems to make sense to place them after the OUTPUT statement when processing multiple records per case.

```sql
label
    alltrans = 'total number of transactions for groceries'
    totalamt = 'total amount spent on groceries'
    aprtrans = 'number of transactions for groceries in April'
    apramt = 'amount spent on groceries in April';
format totalamt apramt dollar10.2;
```

A RUN statement causes the DATA step to be executed:

```sql
run;
```

Figure 6 shows a PROC PRINT output for the data set created by the DATA step. A comparison with Figure 3 above shows that we have produced exactly the same result using multiple record processing as we did using array processing.

![Figure 6--PROC PRINT Output for Summary Data from Example Multiple Records Data Set](image)

**USING FUNCTIONS WITH REPEATED MEASURES DATA**

What if you wanted to know the mean time between visits for a unique id or the mean time between visits for all observations in the file? One way to do this is with the DIF and LAG functions. These two queuing functions are used for accessing prior values of a variable. In simple terms, the LAG function stores values in a queue and returns values from prior calls. The queue is created when the LAG function is called and is initialized as a missing value. The DIF function returns the difference between the value of the current observation and its lag. In order to get accurate results when calling both the DIF and LAG functions, it's critical that the data be sorted in the correct order e.g., chronologically.

Let's use these functions to calculate, for each unique ID, both the average time between visits and the average difference in amount between visits. First we sort the data by ID and date, and for the sake of simplicity we will keep only data on grocery visits:

```sql
proc sort data=sugi.example (where=(purpose='groceries'))
    out=example;
    by id transdate;
run;
```

We now create a new data set and generate three new variables: one for the lag of amount spent on groceries, one for the difference between the amount spent at the current visit and the preceding observation, and one to store the number of days between the current visit and the preceding visit. As we did earlier, we will use a BY statement in the data set to create the temporary variable FIRST_byvariable, which we will use later on in the data step.

```sql
proc sort data=sugi.example (where=(purpose='groceries'))
    out=example;
    by id transdate;
run;
```
data lag_difg;
set example;
  by id transdate;
format amount_lag amount_dif dollar10.2;
amount_lag = lag(amount);
amount_dif = dif(amount);
days_dif = dif(transdate);

If we add a RUN statement and stop here, data for the first eight records in the file are shown in Figure 7.

<table>
<thead>
<tr>
<th>id</th>
<th>purpose</th>
<th>transdate</th>
<th>amount</th>
<th>amount_lag</th>
<th>amount_dif</th>
<th>days_dif</th>
</tr>
</thead>
<tbody>
<tr>
<td>1234</td>
<td>groceries</td>
<td>03/07/2004</td>
<td>$53.45</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>03/15/2004</td>
<td>$54.44</td>
<td>$53.45</td>
<td>$0.99</td>
<td>8</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>03/30/2004</td>
<td>$78.34</td>
<td>$54.44</td>
<td>$23.90</td>
<td>15</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>04/06/2004</td>
<td>$83.34</td>
<td>$78.34</td>
<td>$5.00</td>
<td>7</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>04/13/2004</td>
<td>$77.44</td>
<td>$83.34</td>
<td>$-5.90</td>
<td>7</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>04/22/2004</td>
<td>$88.45</td>
<td>$77.44</td>
<td>$11.01</td>
<td>9</td>
</tr>
<tr>
<td>1235</td>
<td>groceries</td>
<td>04/25/2004</td>
<td>$76.35</td>
<td>$88.45</td>
<td>$-12.10</td>
<td>7</td>
</tr>
<tr>
<td>1235</td>
<td>groceries</td>
<td>03/09/2004</td>
<td>$63.45</td>
<td>$76.35</td>
<td>$-12.90</td>
<td>-57</td>
</tr>
</tbody>
</table>

Figure 7—Results of LAG and DIF Functions

But note that the calculated values for the first occurrence of ID 1234 actually reflect the data from the last occurrence of ID 1234, producing erroneous results. To correct this problem, we use the temporary variable FIRST, to set the value of each new occurrence of an ID to missing. Note that this must be done after using the LAG and DIF functions in the data step.

```plaintext
if first.id then do;
  amount_lag=.;
  amount_dif=.;
  days_dif=.;
end;
run;
```

The corrected results are shown in Figure 8.

<table>
<thead>
<tr>
<th>id</th>
<th>purpose</th>
<th>transdate</th>
<th>amount</th>
<th>amount_lag</th>
<th>amount_dif</th>
<th>days_dif</th>
</tr>
</thead>
<tbody>
<tr>
<td>1234</td>
<td>groceries</td>
<td>03/07/2004</td>
<td>$53.45</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>03/15/2004</td>
<td>$54.44</td>
<td>$53.45</td>
<td>$0.99</td>
<td>8</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>03/30/2004</td>
<td>$78.34</td>
<td>$54.44</td>
<td>$23.90</td>
<td>15</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>04/06/2004</td>
<td>$83.34</td>
<td>$78.34</td>
<td>$5.00</td>
<td>7</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>04/13/2004</td>
<td>$77.44</td>
<td>$83.34</td>
<td>$-5.90</td>
<td>7</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>04/22/2004</td>
<td>$88.45</td>
<td>$77.44</td>
<td>$11.01</td>
<td>9</td>
</tr>
<tr>
<td>1234</td>
<td>groceries</td>
<td>04/25/2004</td>
<td>$76.35</td>
<td>$88.45</td>
<td>$-12.10</td>
<td>7</td>
</tr>
<tr>
<td>1235</td>
<td>groceries</td>
<td>03/09/2004</td>
<td>$63.45</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>1235</td>
<td>groceries</td>
<td>03/13/2004</td>
<td>$88.34</td>
<td>$63.45</td>
<td>$24.89</td>
<td>10</td>
</tr>
</tbody>
</table>

Figure 8—Re-initializing the First Record of a By-group After Using LAG and DIF Functions

Now let's use PROC MEANS to calculate the average time between visits for each and the average difference in amount spent.

```plaintext
proc means data=lag_difg mean maxdec=2;
  class id;
  var amount_dif days_dif;
run;
```

Figure 9 shows the resulting output.
CONVERTING ARRAY DATA TO MULTIPLE RECORDS PER CASE

We have seen how we can use different approaches to achieve the same results with array data and multiple records data. From time to time, you may feel that you want to convert your data set from array format to multiple records. For example, if some cases have a large number of entries while others have only a few, the multiple record format may save a lot of storage space. We will now look at how to convert a data set with an array into a data set which has multiple records per case.

We will use the same array data set as in Figure 1 above, with an array size of 11, since a given ID may have as many as 11 transactions. As you will see, this process involves ARRAY and OUTPUT statements in a slightly different way than we have used them up to now. We begin with a DATA statement that includes a KEEP= option, since we only want to keep the four output variables we will be creating.

```sas
   data multiple (keep=id purpose transdate amount);
   set sugi.vector;
```

We now use ARRAY statements, as we did before, to simplify processing:

```sas
   array pur {*} purpose1-purpose11;
   array trans {*} transdate1-transdate11;
   array amt {*} amount1-amount11;
```

Next we want to combine array processing with an explicit OUTPUT statement, so that we generate multiple records. We accomplish this by including the OUTPUT record in a DO loop. We also use an IF statement to make sure we only write a new record if the array elements are not missing. The SAS code looks like this:

```sas
   do i = 1 to 11;
      if pur {i} ^= '' then do;
         purpose = pur {i};
         transdate = trans {i};
         amount = amt {i};
         output;
      end;
   end;
```

As before, we may add labels and specify formats:

```sas
   label
      id = 'customer id'
      purpose = 'purpose of transaction'
      transdate = 'transaction date'
      amount = 'transaction amount';
   format transdate1-transdate11 mmddyy10.;
   format amount1-amount11 dollar10.2;
```

Finally, a RUN statement tells SAS to execute the DATA step.
The result is a new data set which has a variable number of records per ID, and only the four variables that we created. Figure 10 shows the first two records for this new data set. A comparison with Figure 5 above shows that we have successfully produced the multiple record data set from the array data set.

### Figure 10--Part of a Multiple Record Data Set Converted from Array Data

#### CONVERTING MULTIPLE RECORDS DATA TO ARRAYS

Now let us suppose we want to go the other way, converting from a multiple records per case format into a data set that includes only a single record per case. To do this, we must first determine the size of the arrays we will need in order to store all the data. This requires a simple DATA step, combined with a PROC FREQ, to give us the maximum number of records per case. To simplify matters, we will only keep a single variable, which gives the number of records for each ID. This will involve another use of the by-group processing and RETAIN and OUTPUT statements that we have used above. First, we make sure that the data are sorted by ID, which allows us to use by-group processing:

```plaintext
proc sort data=example;
   by id;
run;
```

Next we define a new DATA step, complete with a BY statement to set up our by-group:

```plaintext
data count (keep=count);
   set multiple;
   by id;
```

We will now define a new variable in order to count the number of records in each by-group. We want to retain its value, so we need a RETAIN statement:

```plaintext
retain count;
```

We initialize this variable when we encounter a new value in our by-group, then increment it each time we encounter a new record within the by-group, and finally write it out when we come to the last record in the by-group:

```plaintext
if first.id then count = 0;
   count + 1;
if last.id then output;
run;
```
Figure 11 shows the output of a PROC FREQ, using the data set we have just created.

### The FREQ Procedure

<table>
<thead>
<tr>
<th>count</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Cumulative Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>8</td>
<td>1</td>
<td>25.00</td>
<td>1</td>
<td>25.00</td>
</tr>
<tr>
<td>9</td>
<td>1</td>
<td>25.00</td>
<td>2</td>
<td>50.00</td>
</tr>
<tr>
<td>10</td>
<td>1</td>
<td>25.00</td>
<td>3</td>
<td>75.00</td>
</tr>
<tr>
<td>11</td>
<td>1</td>
<td>25.00</td>
<td>4</td>
<td>100.00</td>
</tr>
</tbody>
</table>

**Figure 11--Number of Records per Case in Example Data Set**

This tells us that the maximum number of records within the ID by-group is 11, so we need to size our arrays at 11. We are now ready to start creating the new data set. As before, we will use a KEEP= data set option so that our new data set includes only the variables we want.

```plaintext
data vector (keep=id purpose1-purpose11 transdate1-transdate11 amount1-amount11);
set multiple;
by id;
```

We need a LENGTH statement to make sure the variables we will be creating have enough storage space. The LENGTH statement has to come before any other statements.

```plaintext
length purpose1-purpose11 $ 11;
```

Because we need to process multiple records before we write an output record, we need to use a RETAIN statement to specify all the new variables, along with the variable (i) that we will use as the index of the arrays.

```plaintext
retain i purpose1-purpose11 transdate1-transdate11 amount1-amount11;
```

Next we define the arrays, as we have done above.

```plaintext
array pur {*} purpose1-purpose11;
array trans {*} transdate1-transdate11;
array amt {*} amount1-amount11;
```

When we first encounter a new by-group, we must re-initialize all the retained variables.

```plaintext
if first.id then do j = 1 to 11;
pur {j} = '';
trans {j} = .;
amt {j} = .;
end;
```

In addition, we need to re-initialize the indexing variable:

```plaintext
if first.id then i = 0;
```

Each time we encounter a new record within the by-group, we increment the indexing variable:

```plaintext
i + 1;
```

We are now ready to assign values within the arrays we have set up. The indexing variable indicates where in the array each value goes.
if purpose ^= '' then do;
pur {i} = purpose;
trans {i} = transdate;
amt {i} = amount;
end;

When we come to the end of the by-group, we are ready to output the data.

if last.id then output;

Finally, we add any LABEL or FORMAT statements needed, and execute the resulting DATA step.

format transdate1-transdate11 mmddyy10.;
format amount1-amount11 dollar10.2;
run;

Figure 12 shows part of a PROC PRINT output for this newly created data set. As we expected, this new data set has a single record per ID, and separate variables to store data for up to 11 transactions.

<table>
<thead>
<tr>
<th>id</th>
<th>purpose1</th>
<th>transdate1</th>
<th>amount1</th>
<th>purpose2</th>
<th>transdate2</th>
<th>amount2</th>
<th>purpose3</th>
<th>transdate3</th>
<th>amount3</th>
<th>purpose4</th>
<th>transdate4</th>
<th>amount4</th>
<th>purpose5</th>
<th>transdate5</th>
</tr>
</thead>
<tbody>
<tr>
<td>1234</td>
<td>gasoline</td>
<td>03/05/2004</td>
<td>$20.30</td>
<td>groceries</td>
<td>03/07/2004</td>
<td>$53.45</td>
<td>groceries</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1235</td>
<td>groceries</td>
<td>03/03/2004</td>
<td>$33.45</td>
<td>gasoline</td>
<td>03/05/2004</td>
<td>$24.44</td>
<td>groceries</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1236</td>
<td>groceries</td>
<td>03/02/2004</td>
<td>$32.30</td>
<td>groceries</td>
<td>03/04/2004</td>
<td>$85.45</td>
<td>gasoline</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1237</td>
<td>gasoline</td>
<td>03/01/2004</td>
<td>$93.94</td>
<td>groceries</td>
<td>03/03/2004</td>
<td>$66.99</td>
<td>gasoline</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td>transdate3</td>
<td>amount3</td>
<td>purpose4</td>
<td>transdate4</td>
<td>amount4</td>
<td>purpose5</td>
<td>transdate5</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1234</td>
<td>03/15/2004</td>
<td>$54.44</td>
<td>groceries</td>
<td>03/20/2004</td>
<td>$70.94</td>
<td>gasoline</td>
<td>03/20/2004</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1235</td>
<td>03/12/2004</td>
<td>$32.30</td>
<td>gasoline</td>
<td>03/20/2004</td>
<td>$23.20</td>
<td>groceries</td>
<td>03/20/2004</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1236</td>
<td>03/12/2004</td>
<td>$25.55</td>
<td>groceries</td>
<td>03/27/2004</td>
<td>$90.34</td>
<td>gasoline</td>
<td>03/27/2004</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1237</td>
<td>03/11/2004</td>
<td>$37.38</td>
<td>groceries</td>
<td>03/26/2004</td>
<td>$91.88</td>
<td>groceries</td>
<td>03/26/2004</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>id</td>
<td>amount5</td>
<td>purpose6</td>
<td>transdate6</td>
<td>amount6</td>
<td>purpose7</td>
<td>transdate7</td>
<td>amount7</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1234</td>
<td>$25.30</td>
<td>groceries</td>
<td>04/06/2004</td>
<td>$83.34</td>
<td>groceries</td>
<td>04/13/2004</td>
<td>$77.44</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1235</td>
<td>$93.34</td>
<td>gasoline</td>
<td>04/11/2004</td>
<td>$33.55</td>
<td>groceries</td>
<td>04/13/2004</td>
<td>$98.45</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1236</td>
<td>$37.30</td>
<td>groceries</td>
<td>04/10/2004</td>
<td>$80.44</td>
<td>gasoline</td>
<td>04/12/2004</td>
<td>$35.55</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1237</td>
<td>$38.84</td>
<td>groceries</td>
<td>04/02/2004</td>
<td>$96.88</td>
<td>groceries</td>
<td>04/09/2004</td>
<td>$90.98</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

Figure 12--Part of a PROC PRINT Output for the Array Data Set After Conversion from Multiple Records

CONCLUSION

In this tutorial, we have introduced the concept of repeated measures processing, and illustrated these concepts with sample data and SAS DATA step code. We have seen how to process repeated measures data in both array and multiple records formats within a SAS DATA step, and how to convert multiple records to arrays and vice versa. As we have seen, SAS includes several powerful concepts and statements for processing repeated measures data. Using by-group processing, along with RETAIN, ARRAY, IF, DO, and OUTPUT statements, allows us to easily manipulate repeated measures data regardless of its format.

CONTACT INFORMATION

We value and encourage your comments and questions. You may contact the authors:
Terry Fain  Cyndie Gareleck
RAND Corporation  RAND Corporation
1776 Main Street  1776 Main Street
Santa Monica  CA  90401  Santa Monica  CA  90401
310-393-0411x7804  310-393-0411x7815
fain@rand.org  cyndie@rand.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.
Other brand and product names are trademarks of their respective companies.

1 Repeated measures data imported from other software programs are often represented as arrays because some software lack the ability to process multiple records per case.